Cantilever with Axial Force

The objective in this document is to study the deflection and curvature in a cantilevered beam subjected to both lateral and axial force. As shown in the figure, the axial force is either tension or compression; both cases are considered. The first case involves axial compression, as shown here:



Input values

The following parameter values, given in N and mm, are employed in the subsequent plots. λ is the fraction of the axial force relative to the buckling load. The cross-section is circular, made of steel, with diameter d:

$$\text{values1} = \left\{P \to \lambda \frac{\pi^2 EI}{(2\,L)^2}\right\};$$

$$\text{values2} = \left\{EI \to 200\,000 \frac{\pi\, d^4}{64}\right\};$$

$$\text{values3} = \{L \to 5000,\ d \to 50,\ F \to 100\};$$

For reference, the static displacement due to the lateral force alone is, in mm:

$$\frac{F\,L^3}{3\,EI}\ /.\ \text{values1}\ /.\ \text{values2}\ /.\ \text{values3}\ //\ N$$

which yields: 67.9061

Compression

In the absence of distributed load, a second-order differential equation is an appropriate model, as described in the first several pages of Timoshenko & Gere's book entitled Theory of Elastic Stability. That differential equation essentially states that *M=EI w''*, where the bending moment has contributions from both *F* and *P*:

$$EI\, w'' = F\, L \left(1 - \frac{x}{L}\right) + P\, (w[L] - w[x])$$

Observe that the right-hand side contains the displacement at the tip of the cantilever. That is awkward, because that requires the solution to the differential equation. It can be solved iteratively, but another approach is adopted here. Following the previously mentioned Timoshenko book, the origin of the *x*-axis starts at the tip of the cantilever and runs downwards towards the base. In that case, the differential equation reads (notice minus sign, needed because w(x) comes out negative):

$$EI\, w'' = F\, x - P\, w[x]$$

The boundary conditions are zero displacement at x=0 and zero rotation at x=L. That differential equation is here solved using a built-in function in Mathematica:

```
sol = DSolve[{EI w''[x] == F x - P w[x], w[0] == 0, w'[L] == 0}, w, x];
```

The tip displacement is:

```
wTip = w[x] /. sol[[1]] /. x -> L;
```

The value of that displacement for some λ-values are:

```
-wTip /. sol /. values1 /. values2 /. values3 /. λ -> {0.01, 0.25, 0.5, 0.8}
```

which yields: `{{68.5831, 90.2389, 134.881, 335.686}}`

Compare those tip displacements with the solution obtained via matrix structural analysis, including both the "Big P-delta" and "small P-delta" effects. This is here done by applying static condensation to remove the rotational degree of freedom at the tip, without locking it:

```
K11 = 12 EI/L^3 - 6 P/(5 L);
K12 = -6 EI/L^2 + P/10;
K22 = 4 EI/L - 2 P L/15;

Kcond = K11 - K12^2/K22;

F/Kcond /. values1 /. values2 /. values3 /. λ -> {0.01, 0.25, 0.5, 0.8}
```

which yields: {68.583, 90.1884, 134.415, 328.119}

A good match is observed, except for very large axial force values. Next, consider a plot of the displacement along the column, remembering that x=0 is the top, i.e., tip of the column:

```
Plot[w[x] - wTip /. sol /. values1 /. values2 /. values3 /.
  λ -> {0.01, 0.25, 0.5, 0.8}, {x, 0, L /. values3}, PlotStyle → Black]
```

which yields:



The bending moment at the base of the column, in kNm, for the case of zero axial load serves as a reference for the next plot:

```
10^-6 F L /. values3 // N
```

which yields: 0.5

The bending moment for the same axial force levels, in kNm, confirms the increased suffering of the column as the axial compressive force increases:

```
M = 10^-6 EI D[D[w[x] /. sol, x], x];

Plot[M /. values1 /. values2 /. values3 /. λ -> {0.01, 0.25, 0.5, 0.8},
 {x, 0, L /. values3}, PlotStyle → Black]
```

which yields:



Notice that the bending moment diagram from the solution to the differential equation can be calculated in two equivalent ways:

```
Mequil = 10^-6 (F x - P w[x]) /. sol;

Msoln = 10^-6 EI D[D[w[x] /. sol, x], x];
```

```
Plot[{Mequil /. values1 /. values2 /. values3 /. λ -> 0.5,
      Msoln /. values1 /. values2 /. values3 /. λ -> 0.5},
      {x, 0, L /. values3},
      PlotStyle → {{Black, Thin}, {Black, Dashed}},
  PlotLegends →
   Placed[
    LineLegend[{"M from equilibrium", "M from differentiating w(x)"}],
    {Left, Top}]]
```

which yields:



Constant Inclined Tension

Now consider a case in which the lateral force, F, is specified as a percentage of the axial force, which is in tension. The result of this setup is that the orientation of the total resultant force remains constant, as shown in this figure:



The differential equation now changes sign of the axial force term, because *P* is now in tension:

```
sol = DSolve[{EI w''[x] == F x + P w[x], w[0] == 0, w'[L] == 0}, w, x];
```

The revised input to accommodate the variation in *F* is:

```
values3 = {L -> 5000, d -> 50, F -> λ 100};
```

Thip displacement serves as a reference point for the subsequent plot:

```
wTip = w[x] /. sol[[1]] /. x -> L;
```

For λ=0.25, that displacement is:

```
Abs[wTip] /. values1 /. values2 /. values3 /. λ -> 0.25
```

which yields: 13.6244

It no longer makes sense to plot solutions for λ=0, because that means zero load. However, we can now increase the load beyond λ=1 because the axial force has a stabilizing effect, in contrast with the compression case. The displacements plotted below shows that the lateral displacement increases when the load is initially applied. However, as the load increases beyond reasonable values, the column straightens and the slope of its tip aligns with the slope of the external load.

```
Plot[w[x] - wTip /. sol /. values1 /. values2 /. values3 /.
  λ -> {0.25, 0.5, 1, 2, 5, 7, 9}, {x, 0, L /. values3}, PlotStyle → Black]
```

which yields:



That straightening of the column, with curvature concentrating at its base, is reflected in the corresponding moment diagrams:

```
Mequil = 10^-6 (F x + P w[x]) /. sol;

Msoln = 10^-6 EI D[D[w[x] /. sol, x], x];
```

```
Plot[
 {Mequil /. values1 /. values2 /. values3 /. λ -> {0.25, 0.5, 1, 2, 5, 7, 9},
      Msoln /. values1 /. values2 /. values3 /.
   λ -> {0.25, 0.5, 1, 2, 5, 7, 9}},
      {x, 0, L /. values3},
      PlotStyle → {{Black, Thin}, {Black, Dashed}},
 PlotLegends →
  Placed[
   LineLegend[{"M from equilibrium", "M from differentiating w(x)"}],
   {Left, Top}]]
```

which yields:



Tapered Beam

Now consider a tapered version of the cantilever, with the diameter, *d*, varying linearly as shown in this figure:



The following revised input accommodates that tapering:

```
values1 = {F -> λ 100, P -> λ (π^2 EI)/(2 L)^2 /. d -> dTip};

values2 = {EI -> 200 000 (π d^4)/64};

values3 = {d -> dTip + (dBase - dTip) x/L};

values4 = {L -> 5000, dTip -> 50};
```

Let the bending stiffness at the *top* of the column be EI_1:

```
sol = DSolve[{(EI /. values2 /. values3 /. values3) w''[x] == F x + P w[x],
    w[0] == 0, w'[L] == 0}, w, x];
```

Thip displacement serves as a reference point for the subsequent plot:

```
wTip = w[x] /. sol[[1]] /. x -> L;
```

At nearly no tapering, that value is, for reference:

```
Abs[wTip] /. sol /. values1 /. values2 /. values3 /. values4 /. λ -> 0.5 /.
   dBase -> 51 /. x -> L /. values4
```

which yields: {21.3006}

Diminishing displacement for different degrees of tapering, at constant force level:

```
Plot[w[x] - wTip /. sol /. values1 /. values2 /. values3 /. values4 /.
   λ -> 0.5 /. dBase -> {51, 60, 70, 80, 90, 100}, {x, 0, L /. values4},
 PlotStyle → Black]
```

which yields:



Bending moment diagram for different degrees of tapering, at constant force level; notice how the bending moment at the base diminishes with increasing cross-section diameter there, but the shape of the diagram changes:

```
Mequil = 10^-6 (F x + P w[x]) /. sol;

Msoln = 10^-6 EI D[D[w[x] /. sol, x], x];
```

```
Plot[{Mequil /. values1 /. values2 /. values3 /. values4 /. λ -> 0.5 /.
    dBase -> {51, 60, 70, 80, 90, 100},
   Msoln /. values1 /. values2 /. values3 /. values4 /. λ -> 0.5 /.
    dBase -> {51, 60, 70, 80, 90, 100},
   Msoln /. values1 /. values2 /. values3 /. values4 /. λ -> 0.5 /.
    dBase -> 200}, {x, 0, L /. values4},
  PlotStyle → {{Black, Thin}, {Black, Dashed}, Red},
  PlotLegends →
   Placed[
    LineLegend[{"M from equilibrium", "M from differentiating w(x)",
      "Large tapering"}], {Left, Top}]]
```

which yields:



In some situations, an increase in the stiffness in a region of a structure may attract more internal force to that region. That is not true at the base of this cantilever, when tapering is considered. However, it is the case further up the column; the moment in the upper half of the cantilevered column can increase substantially due to the increase in cross-section at the bottom, as highlighted by the red line in the figure above. Notice also how an increased force level affects the bending moment diagram:

```
Plot[{Mequil /. values1 /. values2 /. values3 /. values4 /. λ -> 3.0 /.
    dBase -> {51, 60, 70, 80, 90, 100},
   Msoln /. values1 /. values2 /. values3 /. values4 /. λ -> 3.0 /.
    dBase -> {51, 60, 70, 80, 90, 100},
   Msoln /. values1 /. values2 /. values3 /. values4 /. λ -> 3.0 /.
    dBase -> 200}, {x, 0, L /. values4},
  PlotStyle → {{Black, Thin}, {Black, Dashed}, Red},
  PlotLegends →
   Placed[
    LineLegend[{"M from equilibrium", "M from differentiating w(x)",
      "Large tapering"}], {Left, Top}]]
```

which yields:



Also, keep in mind that the increasing stiffness at the base, from increased tapering, makes the cantilever overall stiffer. In turn, that may attract more lateral force, which could indeed make the moment at the base larger than without tapering.